<?xml version = '1.0'?>
<xsl:stylesheet version="1.0" xmlns:xsl="http://www.w3.org/1999/XSL/Transform">
   <xsl:output method="text"/>
   <xsl:template match="text()"/>
   <!--
	===========	From XGMML   To GML  =================
		   	   by John Punin
-->
   <xsl:template match="*">
      <xsl:value-of select="name()"/> [
<xsl:apply-templates select="@*"/>
      <xsl:apply-templates/>]
</xsl:template>
   <!-- Numeric Attributes -->
   <xsl:template match="@id|@directed|@Scale|@Graphic|@Rootnode        
		 |@source|@target|@x|@y|@z|@w|@h|@d|@width|@smooth
		 |@splinesteps|@extent|@start|@visible">
      <xsl:value-of select="name()"/> <xsl:text>  </xsl:text>
      <xsl:value-of select="."/><xsl:text>
</xsl:text>
</xsl:template>
   <!-- String  Attributes -->
   <xsl:template match="@name|@value|@type|@weight|@labelanchor|@Vendor
                  |@layout|@image|@bitmap|@arrow|@capstyle|@joinstyle
		  |@justify|@font|@style|@stipple|@fill|@outline|@anchor">
      <xsl:value-of select="name()"/> "<xsl:value-of select="."/>"
</xsl:template>
   <!--label-->
   <xsl:template match="@label[contains(.,'&quot;')]">
      <xsl:value-of select="name()"/> "<xsl:value-of select="translate(.,'&quot;',' ')"/>"
</xsl:template>
   <xsl:template match="@label[not(contains(.,'&quot;'))]">
      <xsl:value-of select="name()"/> "<xsl:value-of select="."/>"
</xsl:template>
</xsl:stylesheet>


